F07HUF (CPBCON/ZPBCON) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07HUF (CPBCON/ZPBCON) estimates the condition number of a complex Hermitian positive-definite band matrix A, where A has been factorized by F07HRF (CPBTRF/ZPBTRF).

2 Specification

```
 SUBROUTINE F07HUF(UPLO, N, KD, AB, LDAB, ANORM, RCOND, WORK,
1                  RWORK, INFO)
 ENTRY      cpbcon(UPLO, N, KD, AB, LDAB, ANORM, RCOND, WORK,
1                  RWORK, INFO)
 INTEGER           N, KD, LDAB, INFO
 real              ANORM, RCOND, RWORK(*)
 complex           AB(LDAB,*), WORK(*)
 CHARACTER*1       UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number (in the 1-norm) of a complex Hermitian positive-definite band matrix A:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1.$$

Since A is Hermitian, $\kappa_1(A) = \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty$.

Because $\kappa_1(A)$ is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of $\kappa_1(A)$.

The routine should be preceded by a call to F06UEF to compute $\|A\|_1$ and a call to F07HRF (CPBTRF/ZPBTRF) to compute the Cholesky factorization of A. The routine then uses Higham's implementation of Hager's method [1] to estimate $\|A^{-1}\|_1$.

4 References

[1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates whether A has been factorized as $U^H U$ or LL^H as follows:

if UPLO = 'U', then $A = U^H U$, where U is upper triangular;
if UPLO = 'L', then $A = LL^H$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $N \geq 0$.

3: KD — INTEGER *Input*

On entry: k, the number of super-diagonals or sub-diagonals of the matrix A.

Constraint: KD ≥ 0.

4: AB(LDAB,∗) — ***complex*** array *Input*

Note: the second dimension of the array AB must be at least max(1,N).

On entry: the Cholesky factor of A, as returned by F07HRF (CPBTRF/ZPBTRF).

5: LDAB — INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HUF (CPBCON/ZPBCON) is called.

Constraint: LDAB $\geq$ KD + 1.

6: ANORM — ***real*** *Input*

On entry: the 1-norm of the **original** matrix A, which may be computed by calling F06UEF. ANORM must be computed either **before** calling F07HRF (CPBTRF/ZPBTRF) or else from a copy of the original matrix A.

Constraint: ANORM ≥ 0.0.

7: RCOND — ***real*** *Output*

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, then A is singular to working precision.

8: WORK(∗) — ***complex*** array *Workspace*

Note: the dimension of the array WORK must be at least max(1,2∗N).

9: RWORK(∗) — ***real*** array *Workspace*

Note: the dimension of the array RWORK must be at least max(1,N).

10: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form $Ax = b$; the number is usually 5 and never more than 11. Each solution involves approximately $16nk$ real floating-point operations (assuming $n \gg k$) but takes considerably longer than a call to F07HSF (CPBTRS/ZPBTRS) with 1 right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The real analogue of this routine is F07HGF (SPBCON/DPBCON).

9 Example

To estimate the condition number in the 1-norm (or infinity-norm) of the matrix A, where

$$A = \begin{pmatrix} 9.39 + 0.00i & 1.08 - 1.73i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.08 + 1.73i & 1.69 + 0.00i & -0.04 + 0.29i & 0.00 + 0.00i \\ 0.00 + 0.00i & -0.04 - 0.29i & 2.65 + 0.00i & -0.33 + 2.24i \\ 0.00 + 0.00i & 0.00 + 0.00i & -0.33 - 2.24i & 2.17 + 0.00i \end{pmatrix}.$$

Here A is Hermitian positive-definite, and is treated as a band matrix, which must first be factorized by F07HRF (CPBTRF/ZPBTRF). The true condition number in the 1-norm is 153.45.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07HUF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX, LDAB
      PARAMETER        (NMAX=8,KDMAX=8,LDAB=KDMAX+1)
*     .. Local Scalars ..
      real             ANORM, RCOND
      INTEGER          I, INFO, J, KD, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX), WORK(2*NMAX)
      real             RWORK(NMAX)
*     .. External Functions ..
      real             F06UEF, X02AJF
      EXTERNAL         F06UEF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         cpbcon, cpbtrf
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07HUF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
*
```

```
*        Compute norm of A
*
         ANORM = F06UEF('1-norm',UPLO,N,KD,AB,LDAB,RWORK)
*
*        Factorize A
*
         CALL cpbtrf(UPLO,N,KD,AB,LDAB,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Estimate condition number
*
            CALL cpbcon(UPLO,N,KD,AB,LDAB,ANORM,RCOND,WORK,RWORK,INFO)
*
            IF (RCOND.GE.X02AJF()) THEN
               WRITE (NOUT,99999) 'Estimate of condition number =',
     +           1.0e0/RCOND
            ELSE
               WRITE (NOUT,*) 'A is singular to working precision'
            END IF
         ELSE
            WRITE (NOUT,*) 'A is not positive-definite'
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END
```

9.2 Program Data

```
F07HUF Example Program Data
  4  1                                                 :Values of N and KD
  'L'                                                  :Value of UPLO
 ( 9.39, 0.00)
 ( 1.08, 1.73) ( 1.69, 0.00)
               (-0.04,-0.29) ( 2.65, 0.00)
                             (-0.33,-2.24) ( 2.17, 0.00)  :End of matrix A
```

9.3 Program Results

```
 F07HUF Example Program Results

 Estimate of condition number =  1.22E+02
```